TARGET    = traZoscopio
VERSION   = 0.1
TEMPLATE  = app
QT       += sql

SOURCES = main.cpp \
    mainwindow.cpp \
    qwebcam.cpp \
    tctracker.cpp \
    trazoscopio-widgets.cpp

HEADERS = \
    include/mainwindow.hpp \
    include/qwebcam.hpp \
    include/tctracker.hpp \
    include/widgets.hpp

INCLUDEPATH += \
    ./include \
    ../libcpcap/include \
    ../cvblobslib/include

linux-g++:INCLUDEPATH += \
    /usr/include/qwt \
    /usr/include/opencv2

win32:INCLUDEPATH += \
    C:/Qwt-6.1.0-rc3/include \
    C:/OpenCV2.1/include/opencv \
    C:/OpenCV2.1/include

win32:LIBS += \
    -LC:/Qwt-6.1.0-rc3/lib -lqwt \
    -L../cvblobslib/construido -lblobs \
    -L../libcpcap/construido -lcpcap-1.0 \
    -LC:/OpenCV2.1/lib -lcv210d \
    -LC:/OpenCV2.1/lib -lcxcore210d

linux-g++:LIBS += \
    -L../libcpcap/construido -lcpcap-1.0 \
    -L../cvblobslib/construidos -lblobs \
    -lopencv_calib3d \
    -lopencv_highgui \
    -lopencv_imgproc \
    -lopencv_core \
    -L/opt/qwt-6.0.1/lib -lqwt \
    -ldl

DEFINES += USE_QOBJECT _CRT_SECURE_NO_WARNINGS _CRT_SECURE_NO_DEPRECATES
unix:DEFINES += LINUX
win32:DEFINES += WIN32

linux-g++:desktop.files = traZoscopio.desktop
linux-g++:icon.files = iconos/trazoscopio.png
linux-g++:desktop.path = /usr/share/applications
linux-g++:icon.path= /usr/share/pixmaps
linux-g++:target.path = /usr/bin

win32:deps.path = C:/trazoscopio-install
win32:target.path = C:/trazoscopio-install

CONFIG(release, debug|release) {
    deps.files += \
        $$[QT_INSTALL_BINS]/QtCore4.dll \
        $$[QT_INSTALL_BINS]/QtGui4.dll \
        $$[QT_INSTALL_BINS]/QtSql4.dll \
        $$[QT_INSTALL_BINS]/QtOpenGL4.dll \
        $$[QT_INSTALL_BINS]/QtSvg4.dll \
        C:/Qwt-6.1.0-rc3/lib/qwt.dll
}

CONFIG(debug, debug|release) {
    deps.files += \
        $$[QT_INSTALL_BINS]/QtCored4.dll \
        $$[QT_INSTALL_BINS]/QtGuid4.dll \
        $$[QT_INSTALL_BINS]/QtSqld4.dll \
        $$[QT_INSTALL_BINS]/QtOpenGLd4.dll \
        $$[QT_INSTALL_BINS]/QtSvg4.dll \
        C:/Qwt-6.1.0-rc3/lib/qwt.dll
}

INSTALLS = target
win32:INSTALLS += deps
linux-g++:INSTALLS += icon desktop

RESOURCES += \
    trazoscopio.qrc

OTHER_FILES += \
    traZoscopio.desktop \
    iconos/trazoscopio.png

win32:LIBS += -lstrmiids -lole32 -loleaut32

MOC_DIR = moc
OBJECTS_DIR = obj
DESTDIR = construido
